

Apêndice A

Gramática EBNF (LL1) da FRANKIE estendida com o tipo UNION

<PROGRAM> := **"program"** <IDENTIFIER> ";" <BLOCK> "."

<BLOCK> :=
[<VAR-DECLARATION-SECTION>]
[<PROCEDURE-DECLARATION-SECTION>]
<COMPOSITE-COMMAND>

<VAR-DECLARATION-SECTION> := (<DECLARATION> ";")+

<DECLARATION> := <DECLARATION-SPECIFIERS> <DECLARATION-LIST> |
"union" <IDENTIFIER> **"begin"** <VAR-DECLARATION-SECTION> **"end"**

<DECLARATION-SPECIFIERS> := <TYPE-SPECIFIER>

<TYPE-SPECIFIER> := (**"boolean"** | **"int"**)

<DECLARATION-LIST> := <IDENTIFIER> ("," <IDENTIFIER>)*

<PROCEDURE-DECLARATION-SECTION> := (<PROCEDURE-DECLARATION> ";")+

<PROCEDURE-DECLARATION> := **"procedure"** <IDENTIFIER> [<FORMAL-PARAMETERS>] ";" <BLOCK>

<FORMAL-PARAMETERS> := "(" <FORMAL-PARAMETERS-SECTION> (";" <FORMAL-PARAMETERS-SECTION>)* ")"

<FORMAL-PARAMETERS-SECTION> := [**"var"**] <DECLARATION-LIST> ":" (<TYPE-SPECIFIER> | **"union"**)

<COMPOSITE-COMMAND> := **"begin"** <COMMAND> (";" <COMMAND>)* **"end"**

<COMMAND> := <IDENTIFIER> <COMMAND-TYPE> | <COMPOSITE-COMMAND> |
<CONDITIONAL-COMMAND> | <ITERATION-COMMAND>

<COMMAND-TYPE> := ["." <IDENTIFIER>] "=" <EXPRESSION> | ["(" <EXPRESSION-LIST> ")"]

<CONDITION-COMMAND> := **"if"** "(" <EXPRESSION> ")" <COMMAND> [**"else"** <COMMAND>]

<ITERATION-COMMAND> := **"while"** "(" <EXPRESSION> ")" <COMMAND>

<EXPRESSION> := <SIMPLE-EXPRESSION> [<RELATION> <SIMPLE-EXPRESSION>]

<RELATION> := "=" | "<" | "<" | "<=" | ">=" | ">"

<SIMPLE-EXPRESSION> := ["+" | "-"] <TERM> (("+" | "-" | **"or"**) <TERM>)*

<TERM> := <FACTOR> (("*" | **"div"** | **"and"**) <FACTOR>)*

<FACTOR> := <VARIABLE> | <NUMBER> | "(" <EXPRESSION> ")" | **"not"** <FACTOR>

<VARIABLE> := <IDENTIFIER> ("." <IDENTIFIER>)*

<EXPRESSION-LIST> := <EXPRESSION> (";" <EXPRESSION>)*

Notação :

()* - repetição uma ou mais vezes
()+ - repetição zero ou mais vezes
[] - opcional