Gramática do Pascal Simplificado (LALG)

<PROGRAM> ::= **“program”** <IDENTIFIER> **“;”** <PROGRAM-BODY> **“.”**

<PROGRAM-BODY> ::= <DECLARATION> **“begin”** <COMMANDS> **“end”**

<DECLARATION> ::= <CONSTANTS-DECLARATION> <VARIABLES-DECLARATION> <PROCEDURES-DECLARATION>

<CONSTANTS-DECLARATION> ::= **“const”** <IDENTIFIER> **“=”** <NUMBER> **“;”** <CONSTANTS-DECLARATION> | λ

<VARIABLES-DECLARATION> ::= **“var”** <VARIABLES> **“:”** <VARIABLES-TYPE> **“;”** <VARIABLES-DECLARATION> | λ
<VARIABLES> ::= <IDENTIFIER> <OTHER-VARIABLES>
<OTHER-VARIABLES> ::= **“,”** <VARIABLES> | λ
<VARIABLES-TYPE> ::= **“real”** | **“integer”**

<PROCEDURES-DECLARATION> ::= **“procedure”** <IDENTIFIER> <PARAMETERS> **“;”**
< PROCEDURE-BODY> <PROCEDURES-DECLARATION> | λ

<PARAMETERS> ::= **“(“** <PARAMETERS-LIST> **“)”** | λ
<PARAMETERS-LIST> ::= <VARIABLES> **“:”** <VARIABLES-TYPE> <OTHER-PARAMETERS>
<OTHER-PARAMETERS> ::= **“;”** <PARAMETERS-LIST> | λ

<PROCEDURE-BODY> ::= < LOCAL-DECLARATIONS> **“begin”** <COMMANDS> **“end”** **“;”**
<LOCAL-DECLARATIONS> ::= <VARIABLES-DECLARATION>

<ARGUMENTS-LIST> ::= **“(“** <ARGUMENTS> **“)”** | λ
<ARGUMENTS> ::= <IDENTIFIER> <OTHER-IDENTIFIERS>
<OTHER-IDENTIFIERS> ::= **“;”** <ARGUMENTS> | λ

<COMMANDS> ::= <COMMAND> **“;”** <COMMANDS> | λ
<COMMAND> ::= **“readln”** **“(“** <VARIABLES> **“)”** |
“writeln” **“(“** <VARIABLES> **“)”** |
“repeat” <COMMANDS> **“until”** <CONDITION> |
“while” **“(”** <CONDITION> **“)”** **“do”** <COMMANDS> |
“if” <CONDITION> **“then”** <COMMAND> <FALSE-CONDITION> |
<IDENTIFIER> **“:=”** <EXPRESSION> |
<IDENTIFIER> <ARGUMENTS-LIST> |
“begin” <COMMANDS> **“end”**

<CONDITION> ::= <EXPRESSION> <RELATION> <EXPRESSION>
<FALSE-CONDITION> ::= **“else”** <COMMAND> | λ

<RELATION> ::= **“=”** | **“<>”** | **“>=”** | **“<=”** | **“>”** | **“<“**
<EXPRESSION> ::= <TERM> <OTHER-TERMS>
<TERM> ::= <UNARY-OPERATOR> <FACTOR> <OTHER-FACTORS>

<UNARY-OPERATOR> ::= **“+”** | **“-”** | λ
<OTHER-TERMS> ::= <SUM-OPERATORS> <TERM> <OTHER-TERMS> | λ
<SUM-OPERATORS> ::= **“+”** | **“-”**
<OTHER-FACTORS> ::= <MULTIPLIER-OPERATORS> <FACTOR> <OTHER-FACTORS> | λ
<MULTIPLIER-OPERATORS> ::= **“*”** | **“/”**
<FACTOR> ::= <IDENTIFIER> | <NUMBER> | **“(“** <EXPRESSION> **“)”**
<NUMBER> ::= <INTEGER-NUMBER> | <REAL-NUMBER>